

Power Analysis for General Linear Models in SAS

SAS offers two different procedures to conduct *a priori* sample size and power analysis. PROC POWER does:

- tests, equivalence tests, and confidence intervals for means
- tests, equivalence tests, and confidence intervals for binomial proportions
- multiple regression
- tests of correlation and partial correlation
- one-way analysis of variance
- rank tests for comparing two survival curves
- logistic regression with binary response
- Wilcoxon-Mann-Whitney (rank-sum) test
- Differences between two repeated measures only

PROC GLMPower adds multi-way factorial ANOVA designs for univariate models, but does not have the capacity to do repeated measures. In addition, there is a stand-alone program installed with SAS called “SAS Power and Sample Size” that offers windows-based input to each of these PROCs. Below are examples of power analyses within regression, ANOVA, and repeated measures ANOVA using PROC POWER.

```
* Turn on SAS graphics, set style that allows plot lines to be distinguished
  on the basis of color, style, or maker;
ODS GRAPHICS ON;
ODS HTML STYLE=htmlbluecml;

/* Options within PLOT for PROC POWER:

X = Effect or N or Power for what goes on X-axis, solving for Y-axis
Min = minimum x-axis value, Max = maximum x-axis value
Key = ByCurve (numbered legend) or ByFeature (style legend) or OnCurves (on plot)
VarY (Color or LineStyle or Panel or Symbol BY Dimension) */

/* Options within MULTREG for PROC POWER:

Model = FIXED for categorical predictors, RANDOM for MVN continuous predictors
Alpha = .05 is default, choose others as you wish

INCLUDE ONLY TWO OF THREE BELOW for N predictors:
  NFullPredictors = # of predictors total in regression
  NReducedPredictors = # of predictors in reduced alternative model
  NTestPredictors = difference in # of predictors between models

SPECIFYING EFFECT SIZE: INCLUDE ONLY TWO OF THREE BELOW RSQ OPTIONS
  RsqFull = one or more R2 values of full model
  RsqReduced = one or more R2 values of reduced model
  RsqDiff = one or more diff in R2 between full and reduced models
OR CHOOSE PartialCorr = one or more partial correlations to be detected


NTotal = Total sample size (put . if you want it to be calculated)
Power = . if you want power to be calculated */

TITLE1 "Power to Detect Range of Partial Correlations";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
  MULTREG Model=random NFullPredictors=4 NTestPredictors=1
    Alpha=.05 PartialCorr= .1 .2 .3 .4 .5 NTotal=100 Power=.;
  PLOT X=N Min=50 Max=200 Key=OnCurves; RUN;
```

The POWER Procedure
Type III F Test in Multiple Regression

Fixed Scenario Elements	
Method	Exact
Model	Random X
Number of Predictors in Full Model	4
Number of Test Predictors	1
Alpha	0.05
Total Sample Size	100

Computed Power		
Index	Partial Corr	Power
1	0.1	0.164
2	0.2	0.506
3	0.3	0.855
4	0.4	0.985
5	0.5	>.999




```

TITLE1 "Power to Detect Range of Partial Correlations";
TITLE2 "Will calculate Ntotal for 80% power at each partial correlation";
PROC POWER;
    MULTREG Model=random NFullPredictors=4 NTestPredictors=1
        Alpha=.05 PartialCorr= .1 .2 .3 .4 .5 NTotal=. Power=.80;
    PLOT X=Effect Min=.1 Max=.5 Key=OnCurves; RUN;

```

The POWER Procedure
Type III F Test in Multiple Regression


Computed N Total			
Index	Partial Corr	Actual Power	N Total
1	0.1	0.800	785
2	0.2	0.800	196
3	0.3	0.800	87
4	0.4	0.802	49
5	0.5	0.814	32



```
TITLE1 "Power to Detect Range of Incremental R2";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
    MULTREG Model=random NFullPredictors=3 NTestPredictors=3
        Alpha=.05 RsqReduced=0 RsqDiff=.05 .10 .15 .20 NTotal=100 Power=.;
    PLOT X=N Min=50 Max=200 Key=OnCurves; RUN;
```

The POWER Procedure Type III F Test in Multiple Regression


Computed Power		
Index	R-square Diff	Power
1	0.05	0.440
2	0.10	0.778
3	0.15	0.937
4	0.20	0.987



```
TITLE1 "Power to Detect Range of Incremental R2";
TITLE2 "Will calculate Ntotal for 80% power at each R2";
PROC POWER;
    MULTREG Model=random NFullPredictors=3 NTestPredictors=3
        Alpha=.05 RsqReduced=0 RsqDiff=.05 .10 .15 .20 NTotal=. Power=.80;
    PLOT X=Effect Min=.05 Max=.30 Key=OnCurves; RUN;
```

The POWER Procedure Type III F Test in Multiple Regression

Computed N Total			
Index	R-square Diff	Actual Power	N Total
1	0.05	0.800	214
2	0.10	0.801	105
3	0.15	0.804	69
4	0.20	0.807	51



```

/* Options within ONEWAYANOVA for PROC POWER:

Test = Contrast for df=1 or Overall for omnibus F
GroupMeans = list mean per group separated by |
Stddev = list common SD (no way to do differently per group)
GroupWeights = (relative sample size of each group)
GroupNs = actual group sample sizes
NPerGroup = sample size per group (use if equal)
Contrast = (contrast codes to specific differences)

NTotal = Total sample size (put . if you want it to be calculated)
Power = . if you want power to be calculated */


TITLE1 "Power to Detect Range of Omnibus Between-Group Mean Differences";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
* 3 groups: Drug, Drug+Caffeine, Placebo, using means from previous research;
* SDs should have been 10.8, 12.4, and 7.1, but it only allows a common SD;
   ONEWAYANOVA   GroupMeans= 11.4 | 16.7 | 4.0 Stddev=10.0 GroupWeights=(1 1 1) (2 2 1)
               Test=Overall Alpha=.05 Ntotal=90 Power=.;
   PLOT X=N Min=30 Max=180 Key=OnCurves; RUN;

```

The POWER Procedure Overall F Test for One-Way ANOVA

Fixed Scenario Elements	
Method	Exact
Alpha	0.05
Group Means	11.4 16.7 4
Standard Deviation	10
Total Sample Size	90

Computed Power				
Index	Weights			Power
1	1	1	1	0.995
2	2	2	1	0.980




```

TITLE1 "Power to Detect Range of Specific Between-Group Mean Differences";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
  * 3 groups: Drug, Drug+Caffeine, Placebo, using means from previous research;
  * SDs should have been 10.8, 12.4, and 7.1, but it only allows a common SD;
    ONEWAYANOVA   GroupMeans= 11.4 | 16.7 | 4.0 Stddev=10.0 GroupWeights=(1 1 1) (2 2 1)
    Test=Contrast Alpha=.05 Ntotal=90 Power=.
    Contrast= (-1 -1 2) (-1 1 0);
    PLOT X=N Min=30 Max=180 Key=OnCurves; RUN;
  
```

The POWER Procedure Single DF Contrast in One-Way ANOVA


Fixed Scenario Elements		Computed Power							
Method	Exact	Index		Contrast		Weights		Power	
Alpha	0.05	1	2	-1	-1	2	1	1	1
Group Means	11.4 16.7 4	2	3	-1	1	0	2	2	1
Standard Deviation	10	3	4	-1	1	0	1	1	1
Total Sample Size	90	4		-1	1	0	2	2	1
Number of Sides	2								
Null Contrast Value	0								



```
TITLE1 "Power to Detect Range of Omnibus Between-Group Mean Differences";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
* 3 groups: Drug, Drug+Caffeine, Placebo, using std mean differences;
  ONEWAYANOVA  GroupMeans= .4 | .6 | 0 Stddev=1 GroupWeights=(1 1 1) (2 2 1)
    Test=Overall Alpha=.05 Ntotal=90 Power=.;
  PLOT X=N Min=30 Max=180 Key=OnCurves; RUN;
```

Fixed Scenario Elements	
Method	Exact
Alpha	0.05
Group Means	0.4 0.6 0
Standard Deviation	1
Total Sample Size	90


Computed Power				
Index	Weights			Power
1	1	1	1	0.538
2	2	2	1	0.431



```
TITLE1 "Power to Detect Range of Specific Between-Group Mean Differences";
TITLE2 "Will calculate power for sample sizes around Ntotal";
PROC POWER;
* 3 groups: Drug, Drug+Caffeine, Placebo, using std mean differences;
  ONEWAYANOVA  GroupMeans= .4 | .6 | 0 Stddev=1 GroupWeights=(1 1 1) (2 2 1)
    Test=Contrast Alpha=.05 Ntotal=90 Power=.;
    Contrast= (-1 -1 2) (-1 1 0);
  PLOT X=N Min=30 Max=180 Key=OnCurves; RUN;
```

Fixed Scenario Elements	
Method	Exact
Alpha	0.05
Group Means	0.4 0.6 0
Standard Deviation	1
Total Sample Size	90
Number of Sides	2
Null Contrast Value	0

Computed Power							
Index	Contrast			Weights			Power
1	-1	-1	2	1	1	1	0.599
2	-1	-1	2	2	2	1	0.467
3	-1	1	0	1	1	1	0.119
4	-1	1	0	2	2	1	0.134



```

/* Options within PAIREDMEANS for PROC POWER:

TEST = Diff provides test of mean difference
PairedMeans = Means to be tested for difference
PairedStddevs = Measure-specific SD for means to be tested
Stddev = Common SD for means to be tested
MeanDiff = Mean difference to be tested
Corr = correlations between measures



NPairs = Total # persons measured repeatedly (put . if you want it to be calculated)
Power = power level (put . if you want power to be calculated) */

TITLE1 "Power to Detect Range of Repeated Measures Mean Differences";
TITLE2 "Will calculate power for sample sizes around Npairs";
PROC POWER;
    PAIREDMEANS    MeanDiff= .2 .4 .6 Stddev=1 Corr= .3 .7
                  Alpha=.05 Npairs=25 Power=.;
    PLOT X=N Min=10 Max=50 Key=OnCurves VarY(PANEL BY Corr); RUN;
    
```

The POWER Procedure Paired t Test for Mean Difference

Fixed Scenario Elements	
Distribution	Normal
Method	Exact
Alpha	0.05
Standard Deviation	1
Number of Pairs	25
Number of Sides	2
Null Difference	0

Computed Power			
Index	Mean Diff	Corr	Power
1	0.2	0.3	0.128
2	0.2	0.7	0.236
3	0.4	0.3	0.368
4	0.4	0.7	0.698
5	0.6	0.3	0.682
6	0.6	0.7	0.960



```

OPTIONS nocenter nodate nonumber linesize=MAX pagesize=MAX formdlm='- '
nomprint nosymbolgen nomlogic; TITLE; ODS TRACE OFF; ODS GRAPHICS OFF;

* GENERATE DATA FOR ANOVA MODEL;

* SIMULATION SPECIFICS;
%GLOBAL n1 n2 n3 mean1 mean2 mean3 sd1 sd2 sd3 ngroups nreps SeedVar;
%LET ngroups = 3;
%LET mean1 = 11.4;
%LET mean2 = 16.7;
%LET mean3 = 4.0;
%LET sd1 = 10.0;
%LET sd2 = 10.0;
%LET sd3 = 10.0;
%LET n1 = 20;
%LET n2 = 20;
%LET n3 = 20;
%LET nreps = 100;
%LET SeedVar = 7;

* Generate nreps datasets of fake people;
%MACRO simulateANOVA;
DATA ANOVAsim;
    CALL streaminit(&SeedVar.); *set random seed;
    %DO replication = 1 %TO &nreps.;
        %DO group = 1 %TO &ngroups.;
            %DO n = 1 %TO &n&group.;
                Y = RAND('NORMAL',%SYSEVALF(&mean&group.),%SYSEVALF(&sd&group.));
                mean = &mean&group.; sd = &sd&group.; group=&group.;
                person=&n.; replication=&replication.;
                OUTPUT;
            %END; *person loop;
        %END; *group loop;
    %END; *replication loop;
RUN;
%MEND simulateANOVA;
%simulateANOVA;

* Run ANOVA per dataset, save results;
PROC GLM DATA=ANOVAsim;
    BY replication;
    CLASS group;
    MODEL y = group / SS3;
    ODS OUTPUT FitStatistics=R2 ModelANOVA=SumSquares OverallANOVA=ANOVA LSMeans=Means;
    LSMEANS group;
RUN;

* Code Omnibus F-test as significant or not;
DATA SumSquares; SET SumSquares;
    IF ProbF LT .05 THEN Signif=1; ELSE Signif=0;
RUN;

TITLE "Proportion Significant Omnibus F-test for Group";
PROC MEANS DATA=SumSquares; VAR Signif; RUN;

TITLE "Simulated Group Means";
PROC MEANS DATA=Means; CLASS Group; VAR yLSmean; RUN;
TITLE;

```



```
* SDs should have been 10.8, 12.4, and 7.1, but it only allows a common SD;
%GLOBAL n1 n2 n3 mean1 mean2 mean3 sd1 sd2 sd3 ngroups nreps SeedVar;
%LET ngroups = 3;
%LET mean1 = 11.4;
%LET mean2 = 16.7;
%LET mean3 = 4.0;
%LET sd1 = 10.8;
%LET sd2 = 12.4;
%LET sd3 = 7.1;
%LET n1 = 20;
%LET n2 = 20;
%LET n3 = 20;
%LET nreps = 100;
%LET SeedVar = 77;
%simulateANOVA;
```

```
* FOR UNEQUAL VARIANCES;
PROC MIXED DATA=ANOVAsim METHOD=ML;
    BY replication;
    CLASS group person;
    MODEL y = group / SOLUTION DDFM=KR;
    LSMEANS group / PDIFF=ALL;
    REPEATED / SUBJECT=person*group GROUP=group;
    ODS OUTPUT Diffs=LSMEANDIFF Tests3 = Tests3;
RUN;

DATA Tests3; SET Tests3;
    IF ProbF LT .05 THEN Signif=1; ELSE Signif=0;
RUN;
DATA LSMEANDIFF; SET LSMEANDIFF;
    IF ProbT LT .05 THEN Signif=1; ELSE Signif=0;
RUN;
PROC SORT DATA=LSMEANDIFF; BY Group _Group; RUN;

TITLE "Proportion Significant Omnibus F-test for Group";
PROC MEANS DATA=Tests3 MEAN; VAR Signif; RUN;

TITLE "Simulated Group Mean Differences";
PROC MEANS DATA=LSMEANDIFF MEAN; BY Group _Group; VAR Signif; RUN;
TITLE;

* To clear results viewer in between runs;
ODS HTML CLOSE; ODS HTML;
```